/*/{Protheus.doc} SYNCPART
E a User Function utilizada no lancamento padrao para buscar o CNPJ do cliente/fornecedor
@author Elton Teodoro Alves
@since 14/11/2014
@version P11 R5
@Param cAlias,Caracter,Nome do Alias do cabecalho de notas (SF1 ou SF2)
@Return Caracter String CNPJ do participante
/*/
#Include 'Protheus.ch'

User Function SYNCPART(cAlias)
	
	Local	aArea	:=	GetArea()
	Local	cRet	:=	''
	Local	cLoja	:=	''
	Local	cForCli	:=	''	
	
	If cAlias == 'SF1'
		
		cForCli	:=	'F1_FORNECE'
		cLoja	:=	'F1_LOJA'
		
	ElseIf cAlias == 'SF2'
		
		cForCli	:=	'F2_CLIENTE'
		cLoja	:=	'F2_LOJA'
		
	End If	
	
	If (cAlias == 'SF1' .And. (cAlias)->F1_TIPO $ ('DB')) .Or. (cAlias == 'SF2' .And. .Not. (cAlias)->F2_TIPO $ ('DB'))
		
		DbSelectArea('SA1')
		DbSetOrder(1)
		
		If DbSeek(xFilial('SA1') + (cAlias)->&cForCli + (cAlias)->&cLoja)
			
			If SA1->A1_XPARTIC
				
				cRet := SA1->A1_CGC
				
			End If
			
		End If
		
	ElseIf (cAlias == 'SF2' .And. (cAlias)->F2_TIPO $ ('DB')) .Or. (cAlias == 'SF1' .And. .Not. (cAlias)->F1_TIPO $ ('DB'))
		
		DbSelectArea('SA2')
		DbSetOrder(1)
		
		If DbSeek(xFilial('SA2') + (cAlias)->&cForCli + (cAlias)->&cLoja)
			
			If SA2->A2_XPARTIC
				
				cRet := SA2->A2_CGC
				
			End If
			
		End If
		
	End If
	
Return cRet

